BUILDING THE 'series_op' PROCEDURE INTERACTIVELY

First define the series as in Abramowitz and Stegun

```
> s[1] := 1 + a[1]*x + a[2]*x^2 + a[3]*x^3;
```

$$s_1 := 1 + a_1\, x + a_2\, x^2 + a_3\, x^3$$

```
> s[2] := 1 + b[1]*x + b[2]*x^2 + b[3]*x^3;
```

$$s_2 := 1 + b_1\, x + b_2\, x^2 + b_3\, x^3$$

```
> s[3] := 1 + c[1]*x + c[2]*x^2 + c[3]*x^3;
```

$$s_3 := 1 + c_1\, x + c_2\, x^2 + c_3\, x^3$$

Define a set of unknowns (the coefficients of s3)

```
> unknowns := {c[1],c[2],c[3]};
```

$$unknowns := \{c_1, c_2, c_3\}$$

Define a 'shorthand' procedure for converting P to a polynomial

```
> P := proc(x) convert(x,polynom) end;
```

```
P := proc(x) convert(x,polynom) end
```

Define a specific series to re-express

```
> series_in := 1 / s[1];
```

$$series\_in := \frac{1}{1 + a_1 x + a_2 x^2 + a_3 x^3}$$

Perform a series expansion to high enough order

> series(" , x = 0, 4);

$$1 - a_1 x + \left(-a_2 + a_1^{\,2}\right) x^2 + \left(-a_3 + a_1 a_2 + \left(a_2 - a_1^{\,2}\right) a_1\right) x^3 + \mathrm{O}(x^4)$$

Convert the power series to a polynomial

> p1 := P(");

$$p1 := 1 - a_1 x + \left(-a_2 + a_1^{\,2}\right) x^2 + \left(-a_3 + a_1 a_2 + \left(a_2 - a_1^{\,2}\right) a_1\right) x^3$$

Convert s[3] to a polynomial

> p2 := P(s[3]);

$$p2 := 1 + c_1 x + c_2 x^2 + c_3 x^3$$

Subtract the two converted series (equivalent to equating them)

> p2 - p1;

$$c_1 x + c_2 x^2 + c_3 x^3 + a_1 x - \left(-a_2 + a_1^{\,2}\right) x^2 - \left(-a_3 + a_1 a_2 + \left(a_2 - a_1^{\,2}\right) a_1\right) x^3$$

Extract the coefficients with respect to x

> coeffs(",x);

$$a_2 - a_1^2 + c_2,\ a_1 + c_1,\ c_3 + a_3 - a_1 a_2 - \left(a_2 - a_1^2\right) a_1$$

Convert the coefficient sequence to a set.
Order by order the coefficients must vanish, and Maple assumes "= 0" if there is no "=" in an equation

> {"};

$$\left\{a_2 - a_1^2 + c_2,\ a_1 + c_1,\ c_3 + a_3 - a_1 a_2 - \left(a_2 - a_1^2\right) a_1\right\}$$

Solve the set of equations for c[1], c[2], c[3]

> solve(",unknowns);

$$\left\{c_1 = -a_1,\ c_2 = -a_2 + a_1^2,\ c_3 = -a_3 + 2\, a_1 a_2 - a_1^3\right\}$$

>